

CENTRO DE EDUCAÇÃO PROFISSIONAL DE TIMBÓ – CEDUP TIMBÓ
CURSO TÉCNICO EM INFORMÁTICA COM HABILITAÇÃO EM
DESENVOLVIMENTO DE SOFTWARE – I MÓDULO – 2013/1
DISCIPLINA DE LÓGICA DE PROGRAMAÇÃO
PROFESSOR DOUGLAS ROPELATO

ALGORITMO EM VISUALG PARA CÁLCULO DE CONDUTORES ELÉTRICOS
DE ENTRADA E DISJUNTORES

Roberto Luiz Debarba

RESUMO

O algoritmo apresentado nesse artigo tem como objetivo a execução do cálculo de componentes elétricos de entrada em residências e edifícios de pequeno e médio porte, apresentando informações básicas de potência de diversos itens, tais como tomadas de uso específico, iluminação, cômodos e soma total da potência elétrica da casa. Para tal cálculo foi utilizada como base a norma técnica de instalações elétricas ABNT NBR 5410, o livro de consulta técnica Instalações Elétricas - 15ª Ed. 2013 - CREDER, Helio - Editora Ltc. e normas fornecidas pela CELESC. Ao inserir dados como: quantidade de cômodos presentes no edifício, tensão utilizada, nome, área e perímetro de cada cômodo e quantidade de equipamentos elétricos específicos (chuveiro, forno elétrico, aquecedor de água/ar, entre outros) o algoritmo executa várias operações de forma sequencial para calcular a potência elétrica (W) de cada cômodo, potência de iluminação indicada, potência de tomadas de uso geral (tugs), potência elétrica total dos equipamentos específicos, potência elétrica total da casa e, com base nos presentes valores adquiridos, informa a espessura indicada para os condutores elétricos de entrada e o modelo de disjuntor necessário para manter toda a rede em perfeito funcionamento. Tal algoritmo tem por objetivo proporcionar maior velocidade ao efetuar cálculos no desenvolvimento de projetos, como o de uma nova residência ou edifício, podendo ser utilizado tanto por engenheiros, quanto por pessoas não ligadas essencialmente ao desenvolvimento. Tal ferramenta também pode auxiliar pessoas leigas a conferir se os presentes equipamentos instalados em suas residências atendem os requisitos recomendados pela CELESC.

Palavras-Chave: Elétrica; Predial; Condutores.

1 INTRODUÇÃO

Esse algoritmo tem por objetivo otimizar, de forma simples e objetiva, o cálculo dos principais componentes elétricos de entrada em uma residência ou edifício de pequeno ou médio porte. Ao inserir dados básicos como quantidade de cômodos, área, perímetro e principais equipamentos elétricos presentes no local, é executada de forma lógica e precisa uma série de fórmulas que permitem a obtenção, ao operador do sistema, de dados como potência elétrica de cômodos, tomadas de uso geral, específico, iluminação, potência geral e espessura dos condutores de entrada junto ao tipo e corrente de disjuntor indicado.

O algoritmo foi desenvolvido visando atingir todo o mercado de construção civil predial, desde projetistas e engenheiros, otimizando seu trabalho, até operários que tem como dever executar a obra, podendo assim, projetar por conta própria ou, por seus devidos motivos, verificar se os equipamentos indicados ou já instalados atendem as necessidades. Também tem aplicação ao consumidor contratante dos serviços de construção, podendo, da mesma forma, verificar se toda a rede elétrica de sua residência atende as normas estipuladas pela comunidade acadêmica da área e pela CELESC, proporcionando para si mais segurança e tranquilidade.

Outra aplicação ao algoritmo em estudo é a verificação para atualização da rede em uma residência antiga onde, com o passar do tempo e os ajustes e melhorias das normas técnicas, ou aquisição de novos equipamentos e ampliação da residência, é necessária a troca ou atualização de sua rede elétrica.

2 ESTUDO DE CASO

Na construção civil, muitas vezes existem dúvidas sobre os tipos de componentes elétricos a serem utilizados numa casa, a iluminação de cada cômodo e o material para instalação elétrica correto a ser aplicado na obra. Tendo como exemplo uma casa genérica, um eletricitista precisa calcular três itens básicos para obter a potência elétrica consumida pela residência. Primeiro, ele deve chegar a potência de iluminação adequada para cada cômodo, considerando sua área, de acordo com a norma ABNT NBR 5410, tópico 9.5.2.1.2. Deve-se aplicar 100VA para os primeiros 6m² e 60VA para cada 4m² inteiros restantes. Posteriormente, o profissional necessita chegar a quantidade de tomadas para uso geral (TUGs), segundo a fórmula estipulada no tópico 9.5.2.2.2 da mesma norma, considerando o tipo de cômodo. Para cômodos como sala, quarto, hall, corredor e semelhantes, é aplicada 1 tomadas de 100VA para cada 5m de perímetro ou fração, já em cozinhas, áreas de serviço, banheiros e afins, utiliza-se 1 tomada para cada 3,5m de perímetro ou fração, sendo as três primeiras de 600VA e as demais de 100VA. Por último, é necessário chegar ao número de tomadas para uso específico presentes na casa (TUEs), segundo a necessidade do cliente, e considerar qual será sua utilização (chuveiro, aquecedor de ambiente, ar condicionado, forno elétrico, etc.), para com isso calcular a potência necessária em cada ponto. Feita a soma de todas as TUEs, deve-se aplicar o fator de demanda, valor que varia conforme o resultado obtido anteriormente. A tabela de consumo pode ser encontrada no livro de consulta técnica Instalações Elétricas - 15ª Ed. 2013 - CREDER, Helio - Editora Ltc. Com todos estes dados em mãos, o eletricitista poderá chegar a potência energética total consumida pela casa, onde é feita a soma seguindo a fórmula ($[(VA_{iluminação} * 1) + (VA_{tues} * 0,8)] * \text{fator de demanda} + W_{\text{tomadas específicas}}$) e com tal dado estipular o tipo e corrente de disjuntor indicado, onde com potência elétrica geral da casa menor ou igual a 15000W – monofásico, maior que 15000W e menor ou igual a 25000W – bifásico e maior que 25000W até 75000W – trifásico. Caso a tensão ultrapasse esse valor, deve ser feita a instalação de um transformador.

Sabendo o tipo de disjuntor, divide-se a tensão por 2 para sistemas bifásicos e por 3 para sistemas trifásicos, em seguida dividindo-a novamente pela tensão da rede elétrica, conseguindo assim, a corrente de entrada. Com base no último dado, implanta-se um disjuntor de corrente diretamente maior a corrente adquirida. Por fim, com consulta à tabela disponível no livro técnico anteriormente citado, ou em normas da CELESC, é possível checar a espessura de condutores elétricos de entrada indicadas, sendo no mínimo 10mm².


2.1 REQUISITOS FUNCIONAIS

- O algoritmo deve requisitar a inserção de dados sobre a rede elétrica que alimenta os sistemas da casa ou edifício;
- O algoritmo deve requisitar o número e nome dos cômodos presentes no ambiente;
- O algoritmo deve requisitar a área e perímetro de cada ambiente inserido;
- O algoritmo deve requisitar quantidade e tipo de equipamentos específicos presentes na residência;
- O algoritmo deve, com base nos dados inseridos, realizar os cálculos necessários para chegar aos resultados esperados;
- O algoritmo deve apresentar ao final dos cálculos uma tabela contendo nome, área, potência elétrica de iluminação, potência elétrica das tomadas de uso geral e potência total de cada cômodo.
- O algoritmo deve apresentar a potência elétrica total utilizada pelos equipamentos de uso específico;
- O algoritmo deve apresentar a potência elétrica total utilizada pela casa;
- O algoritmo deve apresentar a espessura indicada dos condutores elétricos de entrada;
- O algoritmo deve apresentar o modelo e corrente de disjuntor ideal para utilização na casa ou residência.


2.3 REQUISITOS NÃO FUNCIONAIS

- O algoritmo deve ser executado no software de aprendizado VisuAlg;
- O algoritmo deve apresentar todas as informações de forma clara e objetiva, para que pessoas com pouco conhecimento na área possam compreendê-las;
- O algoritmo deve apresentar todos os resultados de forma precisa.


3 FLUXOGRAMA




Cálculo de potência de iluminação




Cálculo de tomadas de uso geral



Cálculo de tomadas de uso específico



Exibição dos resultados



4 ALGORITMO

```
algoritmo "Algoritmo em VisuAlg para cálculo de condutores elétricos de entrada e
disjuntores"
// Função : Calcular condutores elétricos de entrada e disjuntores.
// Autor : Roberto Luiz Debarba
// Data : 09/07/2013
//Informações=====
procedimento informacoes
inicio
    escreval ("INFORMAÇÕES")
    escreval ("-----")
    escreval
    escreva ("Digite o número de comodos na casa (max. 20): ")
    leia (numcomodos)
    escreva ("Digite a tensão elétrica da casa (127 ou 220): ")
    leia (tensao)
    calc <- 0
    limpatela
    escreval ("INFORMAÇÕES")
    escreval ("-----")
    escreval
    repita
        calc <- calc + 1
        escreval ("Comodo ", calc)
        escreva ("Digite o nome do comodo: ")
        leia (comodonome[calc])
        comodo[calc,2] <- 0
        se (comodonome[calc] = "Sala") ou (comodonome[calc] = "Dispensa") ou
(comodonome[calc] = "Quarto") ou (comodonome[calc] = "Hall") ou
(comodonome[calc] = "Corredor") ou (comodonome[calc] = "Closet") entao
            comodo[calc,2] <- 1
        fimse
        se (comodonome[calc] = "Copa") ou (comodonome[calc] = "Cozinha") ou
(comodonome[calc] = "Lavanderia") ou (comodonome[calc] = "Lavação") ou
(comodonome[calc] = "Área de serviço") ou (comodonome[calc] = "Banheiro") entao
            comodo[calc,2] <- 2
        fimse
        se (comodo[calc,2] = 0) entao
            escreval ("Tipo de comodo não identificado!")
            escreval ("1 - Sala, Quarto, Hall, Corredor, Closet, semelhantes;")
            escreval ("2 - Copa, Cozinha, Lavanderia, Banheiro, semelhantes.")
            escreva ("Digite o número do grupo no qual o comodo digitado pertence: ")
            leia (comodo[calc,2])
            se (comodo[calc,2] <> 1) e (comodo[calc,2] <> 2) entao
                repita
                    escreval ("Número inválido!")
                    escreva ("Digite novamente: ")
                    leia (comodo[calc,2])
                ate (comodo[calc,2] = 1) ou (comodo[calc,2] = 2)
            fimse
        fimse
        escreva ("Digite a área do comodo (m²): ")
        leia (comodo[calc,1])
        escreva ("Digite o perimetro do comodo (m): ")
        leia (comodo[calc,6])
    escreval
```



```

    ate calc = numcomodos
fimprocedimento
//FIMPROCEDIMENTO=====
//Iluminação=====
procedimento iluminacao
var
    resto : inteiro
inicio
    limpatela
    calc <- 0
    repita
        calc <- calc + 1
        resto <- (int((comodo[calc,1] - 6) / 4))* 60
        comodo[calc,3] <- 100 + resto
    ate calc = numcomodos
fimprocedimento
//FIMPROCEDIMENTO=====
//TUGS=====
procedimento tugs
var
    num1, totaltomadas, sobra, num2 : real
inicio
    calc <- 0
    repita
        limpatela
        calc <- calc + 1
        se (comodo[calc,2] = 1) entao
            num1 <- (int(comodo[calc,6] / 5))
            num2 <- ((comodo[calc,6] / 5) - num1)
            se (num2 <> 0) entao
                num2 <- 1
            senao
                num2 <- 0
            fimse
            comodo[calc,4] <- (num1 + num2) * 100
        senao
            num1 <- (int(comodo[calc,6] / 3.5))
            num2 <- ((comodo[calc,6] / 3.5) - num1)
            se (num2 <> 0) entao
                num2 <- 1
            senao
                num2 <- 0
            fimse
            totaltomadas <- num1 + num2
            se (totaltomadas <= 3) entao
                comodo[calc,4] <- totaltomadas * 600
            senao
                sobra <- (totaltomadas - 3) * 100
                comodo[calc,4] <- (1800 + sobra)
            fimse
        fimse
    ate calc = numcomodos
fimprocedimento
//FIMPROCEDIMENTO=====
//tues=====
procedimento tues
var

```

```

aquecedor, paquecedor, chuveiro, pchuveiro, ar, par, forno, pforno, numtues, agua,
pagua : inteiro
demanda : real
inicio
  limpatela
  escreval ("INFORMAÇÕES")
  escreval ("-----")
  escreval
  escreval ("Digite a quantidade dos seguintes aparelhos elétricos presentes na
casa...")
  escreva ("Aquecedores de Ar: ")
  leia (aquecedor)
  paquecedor <- aquecedor * 1300
  escreva ("Chuveiros Elétricos: ")
  leia (chuveiro)
  pchuveiro <- chuveiro * 7000
  escreva ("Ar Condicionado: ")
  leia (ar)
  par <- ar * 1300
  escreva ("Forno Elétrico: ")
  leia (forno)
  pforno <- forno * 1200
  escreva ("Aquecedor de Água: ")
  leia (agua)
  pagua <- agua * 5000
  numtues <- aquecedor + chuveiro + ar + forno + agua
  escolha numtues
    caso 1
      demanda <- 1
      .
      .
      .
    caso 16
      demanda <- 0.43
    outrocaso
      demanda <- 0.4
  fimescolha
  pespecificos <- (paquecedor + pchuveiro + par + pforno + pagua) * demanda
  limpatela
  escreval ("INFORMAÇÕES")
  escreval ("-----")
  escreval
  escreval ("Informações inseridas com sucesso!")
  escreval ("Pressione ENTER para continuar...")
  leia (varinutil)
fimprocedimento
//FIMPROCEDIMENTO=====
//pporcomodo=====
procedimento pporcomodo
inicio
  calc <- 0
  repita
    calc <- calc + 1
    comodo[calc,5] <- comodo[calc,3] + comodo[calc,4]
  ate calc = numcomodos
fimprocedimento
//FIMPROCEDIMENTO=====

```

```

//PotenciaGeral=====
procedimento potenciageral
var
    luz, tug, num1, num2 : real
inicio
    calc <- 0
    tug <- 0
    luz <- 0
    repita
        calc <- calc + 1
        luz <- comodo[calc,3] + luz
        tug <- comodo[calc,4] + tug
    ate calc = numcomodos
    num1 <- luz + (tug * 0.8)
    se (num1 <= 1000) entao
        num2 <- num1 * 0.86
    fimse
    .
    .
    .
    se (num1 > 10000) entao
        num2 <- num1 * 0.24
    fimse
    ptotal <- num2 + pespecificos
fimprocedimento
//FIMPROCEDIMENTO=====
//Disjuntor=====
procedimento disjuntor
inicio
    se (ptotal < 15000) entao
        tipo <- "Monofásico"
        corrente <- ptotal / tensao
    fimse
    se (ptotal >= 15000) e (ptotal < 25000) entao
        tipo <- "Bifásico"
        corrente <- (ptotal / 2) / tensao
    fimse
    se (ptotal >= 25000) e (ptotal < 75000) entao
        tipo <- "Trifásico"
        corrente <- (ptotal / 3) / tensao
    fimse
    se (ptotal >= 75000) entao
        tipo <- "Transformador"
        corrente <- 0
    fimse
    se (corrente > 0) e (corrente < 1) entao
        pdisjuntor <- 1
    fimse
    .
    .
    .
    se (corrente >= 175) e (corrente <= 200) entao
        pdisjuntor <- 200
    fimse
    se (corrente = 0) ou (corrente > 200) entao
        pdisjuntor <- 0
    fimse

```

```

fimprocedimento
//FIMPROCEDIMENTO=====
//Cabos=====
procedimento cabos
inicio
  se (tipo = "Monofásico") entao
    se (pdisjuntor > 0) e (pdisjuntor < 69) entao
      cabo <- 10
    fimse
    .
    .
    .
    se (pdisjuntor >= 201) ou (pdisjuntor = 0) entao
      cabo <- 0
    fimse
  senao //-----Bifásico e Trifásico-----
    se (pdisjuntor > 0) e (pdisjuntor < 62) entao
      cabo <- 10
    fimse
    .
    .
    .
    se (pdisjuntor >= 179) ou (pdisjuntor = 0) entao
      cabo <- 0
    fimse
  fimse
fimprocedimento
//FIMPROCEDIMENTO=====
//Resultados=====
procedimento resultado
inicio
  escreval ("RESULTADO")
  escreval ("-----")
  escreval
  calc <- 0
  escreval ("Comodo / Área / Iluminação / Tugs / Potência Elétrica
Geral")
  escreval
  repita
    calc <- calc + 1
    escreval (comodonome[calc], " ", comodo[calc,1], " m² ",
comodo[calc,3], " W ", comodo[calc,4], " W ", comodo[calc,5], " W")
    ate calc = numcomodos
  escreval
  escreval ("Potência Elétrica das tomadas especificas: ", pespecificos, " W")
  escreval
  escreval ("-----")
  escreval
  escreval ("Potência Elétrica total da casa (Considerado Fator de Demanda): ", ptotal,
" W")
  escreval
  escreval ("-----")
  escreval
  se (cabo <> 0) entao
    escreval ("Espessura de cabos indicada: ", cabo, " mm²")
  senao
    escreval ("Espessura de cabos indicada: ERRO!")

```

```

        escreval ("Corrente muito alta. Espessura indicada não consta em nosso banco de
dados!")
    fimse
    escreval
    se (pdisjuntor <> 0) entao
        escreval ("Disjuntor indicado: ", tipo, " de ", pdisjuntor, " A")
    senao
        escreval ("Disjuntor indicado: ", tipo)
        escreval ("Potência elétrica muito alta!")
        escreval ("A especificação do equipamento não consta em nosso banco de
dados!")
    fimse
fimprocedimento
//FIMPROCEDIMENTO=====
var
    comodo : vetor [1..20,1..6] de real
    comodonome : vetor [1..20] de caractere
    corrente, cabo, pespecificos, numcomodos, tensao, ptotal, opcao, pdisjuntor : real
    calc : inteiro
    varinutil, tipo, continuar : caractere
inicio
    escreval ("Calculadora de condutores de entrada e disjuntores")
    escreval ("-----")
    escreval
    escreval ("|-----Menu-----|")
    escreval ("|")
    escreval ("|          1 - Iniciar          |")
    escreval ("|          0 - Sair            |")
    escreval ("|")
    escreval ("|-----|")
    escreval
    escreval
    escreva ("Digite uma opção: ")
    leia (opcao)
    limpatela
    se (opcao = 1) entao
        continuar <- "s"
        enquanto (continuar = "s") faca
            limpatela
            informacoes
            iluminacao
            tugs
            tues
            potenciageral
            disjuntor
            cabos
            pporcomodo
            limpatela
            resultado
            escreval
            escreva ("Calcular novamente (s/n): ")
            leia (continuar)
        fimenquanto
    fimse
fimalgoritmo

```

5 TESTE DE MESA

MATRIZ comodonome	
	1
1 .. 20	Nome do comodo

MATRIZ comodo						
	1	2	3	4	5	6
1 .. 20	Área	Tipo	Iluminação	Tugs	Potência	Perimetro

Matrizes	
comodonome[1]	Cozinha
comodonome[2]	Quarto
comodo[1,1]	20
comodo[2,1]	10
comodo[1,2]	2
comodo[2,2]	1
comodo[1,3]	280
comodo[2,3]	160
comodo[1,4]	2200
comodo[2,4]	300
comodo[1,5]	2480
comodo[2,5]	460
comodo[1,6]	24
comodo[2,6]	14

Inicio	
corrente	
cabos	
pespecificos	
numcomodos	
tensao	
calc	
ptotal	
opcao	1
pdisjuntor	
varinutil	
tipo	
continuar	s

informacoes	
corrente	
cabos	
pespecificos	
numcomodos	2
tensao	220
calc	0 / 1 / 2
ptotal	
opcao	1
pdisjuntor	
varinutil	-
tipo	
continuar	s

iluminacao			
	Inicial	1º comodo	2º comodo
corrente			
cabos			
pespecificos			
numcomodos	2	2	2
tensao	220	220	220
calc	0	1	2
ptotal			
opcao	1	1	1
pdisjuntor			
varinutil	-	-	-
tipo			
continuar	s	s	s
resto	0	180	60

tugs			
	Inicial	1º cômodo	2º cômodo
corrente			
cabos			
pespecificos			
numcomodos	2	2	2
tensao	220	220	220
calc	0	1	2
ptotal			
opcao	1	1	1
pdisjuntor			
varinutil	-	-	-
tipo			
continuar	s	s	s
num1	-	6	2
totaltomadas	-	7	-
sobra	-	400	-
num2	-	0,85714 / 1	0,8 / 1

tues	
corrente	
cabos	
pespecificos	11020
numcomodos	2
tensao	220
calc	1
ptotal	
opcao	1
pdisjuntor	
varinutil	-
tipo	
continuar	s
aquecedor	0
paquecedor	0
chuveiro	1
pchuveiro	7000
ar	1
par	1300
forno	1
pforro	1200
agua	1
pagua	5000
numtues	4
demanda	0,76

potenciageral			
	Inicial	1º cômodo	2º cômodo
corrente			
cabos			
pespecificos	11020	11020	11020
numcomodos	2	2	2
tensao	220	220	220
calc	0	1	2
ptotal	-	-	12630,4
opcao	1	1	1
pdisjuntor			
varinutil	-	-	-
tipo			
continuar	s	s	s
luz	0	280	440
tug	0	2200	2500
num1	-	-	2440
num2	-	-	1610,4

disjuntor	
corrente	57,4109
cabo	
pespecificos	11020
numcomodos	2
tensao	220
calc	0
ptotal	12630,4
opcao	1
pdisjuntor	60
varinutil	-
tipo	Monofásico
continuar	s

cabos	
corrente	57,4109
cabo	10
pespecificos	11020
numcomodos	2
tensao	220
calc	0
ptotal	12630,4
opcao	1
pdisjuntor	60
varinutil	-
tipo	Monofásico
continuar	s

pporcomodo	
corrente	57,4109
cabo	10
pespecificos	11020
numcomodos	2
tensao	220
calc	0 / 1 / 2
ptotal	12630,4
opcao	1
pdisjuntor	60
varinutil	-
tipo	Monofásico
continuar	s

6 CONSIDERAÇÕES FINAIS

Um dos maiores desafios encontrados no desenvolvimento desse artigo, mesmo parecendo um tanto insignificante, foi a escolha de seu tema. A meta do trabalho, considerada a plataforma onde foi desenvolvido seu algoritmo, se direcionou a elaboração de algum cálculo. Após algum tempo em busca de um tema viável, ao mesmo tempo suficientemente complexo, o fato de contar com o contato diário com vários professores engenheiros eletricitas se mostrou favorável. A escolha por algum cálculo voltado à área foi fácil, porém um novo desafio deveria ser superado, o completo entendimento do tema para então aplicá-lo de forma clara e eficiente ao algoritmo. O apoio de alguns professores foi essencial, facilitando muito tal processo.

Após o desenvolvimento de toda a lógica, que se mostrou bastante simples, apesar de extensa, o ato de convertê-la em algoritmo não apresentou nenhuma dificuldade, considerando novamente a simplicidade da plataforma VisuAlg. Apenas poucas modificações na lógica inicial foram necessárias para simplificá-la, encurtá-la ou até mesmo aplicar algum incremento que se mostrou útil.

Algo que não pode ser desconsiderado é o fato de, considerando que a forma final do trabalho é um artigo científico, o número máximo de páginas limitou o número de funcionalidades possíveis a serem aplicadas, mesmo com a abstração de alguns fluxogramas e estruturas na forma final do algoritmo.

7 REFERÊNCIAS

CREDER, Helio. **Instalações Elétricas**. 15. ed. Rio de Janeiro, RJ: Ltc, 2008.

ABNT - ASSOCIAÇÃO BRASILEIRA DE NORMAS TÉCNICAS (Org.). **ABNT NBR 5410**: Instalações elétricas de baixa tensão. 2. ed. Rio de Janeiro, RJ: Abnt, 2008. Versão corrigida, 2008.

PRYSMIAN ENERGIA CABOS E SISTEMAS DO BRASIL S. A. (Brasil) (Org.). **Instalações Elétricas Residenciais**. Santo Andre, SP: Victory, 2006.